

## การใช้ script มาช่วยในการตัดการเชื่อมต่อ hotspot user ตามจำนวนวันใช้งาน

สำหรับอุปกรณ์ Mikrotik สามารถทำระบบ hotspot ได้ดีมาก แต่ยังมีข้อจำกัดในเรื่องของการตัดการเชื่อมต่อ ตามจำนวนวันใช้งาน โดยเริ่มนับจากการใช้งานครั้งแรกที่ Login ซึ่งปกติจะทำได้ต้องมีระบบ Radius หรือ API หรือไม่ต้องลง package Userman มาช่วยในการกำหนดความสามารถนี้ แต่ผมจะใช้วิธีการทำ script ไปใส่ในอุปกรณ์ Mikrotik เพื่อเช็ควันหมดอายุของผู้ใช้งาน โดยที่ไม่ต้องพึ่ง software หรือ application มาติดตั้งเพิ่ม นอกจาก script ที่จะต้องนำไปใส่ในอุปกรณ์ Mikrotik

### หลักการทำงาน

การทำงานของ script จะแบ่งเป็น 3 ส่วน คือ

1. Script ที่นำไปใส่ใน On Login ของ Hotspot User Profile เพื่อไประบุใน comment ของ Hotspot User ว่าให้หมดอายุกี่วันหลังจากการ Login ครั้งแรก ถ้า Login ครั้งแรกวันที่ 06 Jul 2019 เวลา 23:36:00 และ User Profile นี้ต้องการกำหนดให้ใช้งานได้เพียง 3 วัน Script นี้ก็จะไปบันทึกลงใน Comment ของ Hotspot User เป็น expire=2019-06-09 23:36:00 เพื่อเป็นข้อมูลให้ script อีก script ที่ทำหน้าที่ Remove Active Hotspot User และทำการ Disable Hotspot User ที่ใช้งานเกินเวลา

The screenshot shows the Mikrotik WinBox interface. The top menu bar includes Session, Settings, and Dashboard. The left sidebar has a tree view with categories like Quick Set, CAPsMAN, Interfaces, Wireless, Bridge, PPP, Switch, Mesh, IP, MPLS, Routing, System, Queues, Files, Log, RADIUS, and Tools. The main window is titled 'Hotspot' and contains tabs for Servers, Server Profiles, Users, User Profiles, Active, Hosts, IP Bindings, Service Ports, Walled Garden, Walled Garden IP List, and Cookies. The 'Users' tab is selected, showing a list of users. Below the list, there's a 'Hotspot User Profile <admin>' window with tabs for General, Queue, and Scripts. The 'Scripts' tab is active, showing a script for 'On Login' that sets a global datetime, calculates a shift date, and sets a local date. Below the script window, there's a table of users. The table has columns for Name, Session Time, Idle Timeout, Shared U..., and Rate Limit (p/c/b). The first user is 'mee' with a session time of '04:59:15' and an idle timeout of '455d 15:11...'. The second user is 'admin' with a session time of '04:59:15' and an idle timeout of '455d 15:11...'. Below the table, there's a 'Hotspot User <mee>' window with tabs for General, Limits, and Statistics. The 'General' tab is active, showing fields for Server, Name, Password, Address, MAC Address, Profile, Routes, and Email. The 'Name' field is set to 'mee' and the 'Profile' field is set to 'admin'. Below the 'Hotspot User <mee>' window, there's a 'Comment for Hotspot User <mee>' window showing the comment 'expire=2019-06-06 05:28:41'.

Name	Session Time...	Idle Timeout	Shared U...	Rate Limit (p/c/b)
default				
admin	UU:UU:UU No expire			
mee	04:59:15	455d 15:11...		
		00:00:00		
		1319d 10:0...		
		3d 16:22:52		
		00:00:00		
		00:00:00		
		514d 23:31...		

จากภาพ Script จะไปสร้าง Comment ของ User : mee ให้มีค่า expire=วันเวลาที่หมดอายุ

```

:global datetime

:global expiredatetime

:local shiftDay 3

:local shiftDate [ :parse [ /system script get func_shiftDate source ] ]

:local montharray ( "jan", "feb", "mar", "apr", "may", "jun", "jul", "aug", "sep", "oct", "nov", "dec" )

:local date [ /system clock get date ]

:local date [ $shiftDate date=$date days=$shiftDay ]

:local day [ :pick $date 4 6 ]

:set day ($day)

:local monthtext [ :pick $date 0 3 ]

:local month ( [ :find $montharray $monthtext ] + 1)

:local year [ :pick $date 7 11 ]

if ($month < 10) do={:set month ("0" . $month)};

:local timetext [ /system clock get time ]

:local hour [ :pick $timetext 0 2 ]

:local minute [ pick $timetext 3 5 ]

:local second [ pick $timetext 6 8 ]

:set datetime ( $year . $month . $day . $time . $hour . $minute . $second )

:set expiredatetime ( $year . "-" . $month . "-" . $day . " " . $hour . ":" . $minute . ":" . $second )

:local i [ /ip hotspot user find name=$user]

:local usercomment [ /ip hotspot user get $i comment ]

if ( [ :find $usercomment "expire" ] != 0 ) do={

    [ /ip hotspot user set $i comment=("expire=" . $expiredatetime) ]

}

```

จาก Script จะมีการกำหนดตัวแปร :local shiftDay 3 เลข 3 คือ จำนวนวันที่กำหนดให้หมดอายุ เช่นหากต้องการให้ใช้งานได้เพียงแค่ 1 วัน ก็ต้องไปกำหนดบรรทัด :local shiftDay 3 เป็น :local shiftDay 1 ระบบก็จะคำนวณวันหมดอายุเป็น 1 วันหลังจากมีการ Login ใช้งานครั้งแรก โดยการคำนวณวันเพิ่มนี้จะไปเรียก script ชื่อ func\_shiftDate มาทำหน้าที่ในการคำนวณ

2. Script ที่ทำหน้าที่คำนวณวันหมดอายุ ชื่อ func\_shiftDate ซึ่งจะต้องไปสร้าง Script นี้ไว้ใน /system script ของอุปกรณ์ Mikrotik โดยกำหนด Policy เพียงแค่ read, write, test และชื่อจะต้องเป็น func\_shiftDate เพราะจะถูกเรียกใช้โดย On Login Script

Script <func\_shiftDate>

Name:

Owner:

☐ Don't Require Permissions

Policy: ☐ ftp ☐ reboot  
☒ read ☒ write  
☐ policy ☒ test  
☐ password ☐ sniff  
☐ sensitive ☐ romon  
☐ dude

Last Time Started:

Run Count:

```
##### func_shiftDate - add days to date
# Input: date, days
# date - "jan/1/2017"
# days - number
# correct only for years >1918
##### uncomment for testing
#local date "jan/01/2100"
#local days 2560
#####
#put "$date + $days"
local mdays {31;28;31;30;31;30;31;31;30;31;30;31}
local months {"jan"=1;"feb"=2;"mar"=3;"apr"=4;"may"=5;"jun"=6;"jul"=7;"aug"=8;"sep"=9;"oct"=10;"nov"=11;"dec"=12}
local monthr {"jan";"feb";"mar";"apr";"may";"jun";"jul";"aug";"sep";"oct";"nov";"dec"}
```

สำหรับ Script นี้ก็สามารถ copy ไปวางใน /system script ได้เลย ไม่ต้องมีการแก้ไขใด ๆ ใน Script เพียงแต่ต้องกำหนด Name และ Policy ของ Script

```
##### func_shiftDate - add days to date
# Input: date, days
# date - "jan/1/2017"
# days - number
# correct only for years >1918
##### uncomment for testing
#:local date "jan/01/2100"
#:local days 2560
#####
#:put "$date + $days"
:local mdays {31;28;31;30;31;30;31;31;30;31;30;31}
:local months {"jan"=1;"feb"=2;"mar"=3;"apr"=4;"may"=5;"jun"=6;"jul"=7;"aug"=8;"sep"=9;"oct"=10;"nov"=11;"dec"=12}
:local monthr {"jan";"feb";"mar";"apr";"may";"jun";"jul";"aug";"sep";"oct";"nov";"dec"}

:local dd [:tonum [:pick $date 4 6]]
:local yy [:tonum [:pick $date 7 11]]
:local month [:pick $date 0 3]

:local mm (:$months->$month)
:set dd ($dd+$days)

:local dm [:pick $mdays ($mm-1)]
:if ($mm=2 && (($yy&3=0 && ($yy/100*100 != $yy)) || $yy/400*400=$yy) ) do={ :set dm 29 }

:while ($dd>$dm) do={
  :set dd ($dd-$dm)
  :set mm ($mm+1)
  :if ($mm>12) do={
    :set mm 1
    :set yy ($yy+1)
  }
  :set dm [:pick $mdays ($mm-1)]
  :if ($mm=2 && (($yy&3=0 && ($yy/100*100 != $yy)) || $yy/400*400=$yy) ) do={ :set dm 29 }
};
:local res "$[:pick $monthr ($mm-1)]/"
:if ($dd<10) do={ :set res ($res."0") }
:set $res "$res$dd/$yy"
:return $res
```

3. Script สุดท้ายเป็น Script ที่ทำหน้าที่เช็ค Comment ของ Hotspot User ว่ามีค่า expire เกินกำหนดวันหมดอายุแล้วหรือไม่ ถ้าเช็คแล้ววันเวลาที่หมดอายุเกินวันเวลาปัจจุบัน Script ก็จะสั่ง Remove Hotspot User และทำการ Disable Hotspot User นั้น โดยการใช้งานนี้ต้องมีการไปตั้งค่าใน /system scheduler เพื่อสั่งให้ run script ทุก ๆ ที่นาทีเดียว เพื่อให้มีการเช็ควันเวลาหมดอายุเป็นระยะ ถ้าตั้งน้อยไปตัวอุปกรณ์ Mikrotik ก็อาจจะทำงานหนักเกินไป ถ้าตั้งมากเกินไป Hotspot User ก็จะใช้งานได้เกินเวลาหมดอายุจริงมากเกินไป Script นี้ผมตั้งชื่อให้เป็น auto-cutoff หากจะตั้งเป็นชื่ออื่นก็ตามแต่ชอบครับ

```
{
:global datetime
:global expiredate
:global datenum
:global expirenum

:local montharray ("jan", "feb", "mar", "apr", "may", "jun", "jul", "aug", "sep", "oct", "nov", "dec")
:local date [/system clock get date]
:local day [ :pick $date 4 6 ]
:local monthtext [ :pick $date 0 3 ]
:local month ([ :find $montharray $monthtext ] + 1)
:local year [ :pick $date 7 11 ]
:if ($month < 10) do={set month ("0", $month)};


```

อย่าลืมไปตั้งใน /system scheduler ด้วยนะครับ

Name	Start Date	Start Time	Interval	Owner	Run Count	Next Run
auto-cutoff	Jul/06/2018	04:08:04	00:05:00	admin	101	Jun/06/2019
dyndns-update					9	Jun/06/2019
gettikSerial					1	

```

{
:global datetime
:global expiredate
:global datenum
:global expirenum

:local montharray ( "jan", "feb", "mar", "apr", "may", "jun", "jul", "aug", "sep", "oct", "nov", "dec" )
:local date [ /system clock get date ]
:local day [ :pick $date 4 6 ]
:local monthtext [ :pick $date 0 3 ]
:local month ( [ :find $montharray $monthtext ] + 1)
:local year [ :pick $date 7 11 ]
:if ( $month < 10) do={:set month ("0" . $month)};;
:local timetext [ /system clock get time ]
:local hour [ :pick $timetext 0 2 ]
:local minute [ pick $timetext 3 5 ]
:local second [ pick $timetext 6 8 ]
:set datetime ( $year . $month . $day . $time . $hour . $minute . $second )
:set datenum [ :tonum $datetime ]

:foreach i in [ /ip hotspot user find where disabled=no ] do={
:local usercomment [ /ip hotspot user get $i comment ]
:local username [ /ip hotspot user get $i name ]
:if ( [ :find $usercomment "expire" ] = 0 ) do={
:local expire [ :pick $usercomment 7 26 ]
:local expireyear [ :pick $expire 0 4 ]
:local expiremonth [ :pick $expire 5 7 ]
:local expireday [ :pick $expire 8 10 ]
:local expirehour [ :pick $expire 11 13 ]
:local expireminute [ :pick $expire 14 16 ]
:local expiresecond [ :pick $expire 17 19 ]
:set expiredate ( $expireyear . $expiremonth . $expireday . $expirehour . $expireminute . $expiresecond )
:set expirenum [ :tonum $expiredate ]
:if ( $datenum > $expirenum ) do={
:log info "HOTSPOT EXPIRE SCRIPT: Disabling Hotspot user $name first logged in $date"
:local expirecomment ("This user expire @" . $expireyear . "-" . $expiremonth . "-" . $expireday . " " . $expirehour . ":" .
$expireminute . ":" . $expiresecond )
[ /ip hotspot user disable $i ]
[ /ip hotspot user set $i comment=$expirecomment ]
[ /ip hotspot active remove [ find where user=$username] ]
}
}
}
}

```

## ที่มาของ Script ต่าง ๆ

Script ที่ใช้กำหนดวันหมดอายุใน comment ให้กับ Hotspot User ผมสร้างขึ้นโดยใช้หลักการแปลงวันที่ระบบของ Mikrotik ซึ่งชื่อเดือนจะเป็นตัวอักษร ต้องแปลงให้มาเป็นตัวเลข และเพิ่มจำนวนตามจำนวนวันที่ต้องการให้หมดอายุเข้าไป และไปสร้างค่า expire ใน Comment ของ Hotspot User

Script ที่ใช้คำนวณเพิ่มวันจากวันที่ปัจจุบัน -> <https://forum.mikrotik.com/viewtopic.php?f=9&t=127050>

Script ที่ใช้ตัดการใช้งานของ Hotspot User -> [https://wiki.mikrotik.com/wiki/Expire\\_users\\_a\\_after\\_number\\_of\\_days](https://wiki.mikrotik.com/wiki/Expire_users_a_after_number_of_days)

การใช้งาน Script ให้ทำงานเป็น Function -> <https://wiki.mikrotik.com/wiki/Manual:Scripting#Functions>

สำหรับท่านที่นำไปใช้งานแล้วติดปัญหาสามารถแจ้งกลับได้ที่ LINE ID : aisfttx

บทความนี้สามารถนำไปดัดแปลงใช้งานและแก้ไขหรือเผยแพร่ได้ตามอัธยาศัยครับไม่มีสงวนลิขสิทธิ์